Copyright (C) 2020 Timothy Fossum

Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.1 or any later version published by the Free Software Foundation. A copy of the license is included in the file "COPYING", entitled "GNU Free Documentation License".

Classes and Objects

Compared to most object-oriented programming languages such as Java and C++, our classes and objects are *first class* – that is, they can be created at any time and in any environment, they can be passed as parameters to procedures, and they can be returned as values of procedure applications.

A *class* is an expressed value that captures a collection of variables (called *fields*) and procedures (called *methods*) and that serves as a template to create expressed values called *instances* of the class. The instances are also called *objects*, to distinguish them from classes. You can think of a class as a factory that identifies how to churn out an arbitrary number of instances of the class.

All classes of an OBJ program belong to a *class hierarchy*, which is a forest – a set of trees – with an unnamed class at the root of each of its trees, and with program-created classes at the other nodes of the trees.

In the class hierarchy, a class X that occurs as a child node of class Y in the class hierarchy is called a *subclass* of Y, and Y is called a *superclass* of X.

When an object of class X is instantiated, instances of each of the classes that lie on the path from X to the root of the tree are created, and the combination of all those instances is considered as the resulting object.

If Y is the superclass of X, then an object x created from class X "contains" an object y created from Y. The object y is called the *parent* of x, and likewise x is called the *child* of y.

A class may also have static variables whose values are shared among all instances of the class.

Consider the following example.

```
define c1 =
  class % extends an unnamed class
    field x
    field y
  end
define c2 =
  class extends c1
    field z
    field x
  end
define obj1 = new c1
define obj2 = new c2
\langle obj1 \rangleset x = 3
\langle obj2\rangleset x = 5
<obj1>x % evaluates to 3
```

In this example, c1 is a class and c2 is a subclass of c1. Object obj1 is an instance of c1 and obj2 is an instance of c2.

We define the concrete and abstract syntax of classes and objects. We also add the additional reserved word nil to our language, which represents a separate expressed value not shared by any other data type. When used in conditional expressions, nil is considered false.

```
<exp>:NilExp
                ::= NIL
                NilExp()
<exp>:ClassExp ::= CLASS <ext> <statics> <fields> <methods> END
                ClassExp(Ext ext, Statics statics, Fields fields, Methods methods)
<ext>:Ext0
                ::=
                Ext0()
\langle ext \rangle : Ext 1
                ::= EXTENDS <exp>
                Ext1(Exp exp)
                **= STATIC <VAR> EQUALS <exp>
<statics>
                Statics(List<Token> varList, List<Exp> expList)
<fields>
                **= FTELD <VAR>
                Fields(List<Token> varList)
<methods>
                **= METHOD <VAR> EQUALS <proc>
                Methods(List<Token> varList, List<Proc> procList)
<exp>:NewExp
                ::= NEW <exp>
                NewExp(Exp exp)
<exp>:EnvExp
                ::= LANGLE <exp>vExp RANGLE <exp>eExp
                EnvExp(Exp vExp, Exp eExp)
```

Every Exp expression evaluates to a Val object, and a ClassExp is no exception, so evaluating such a ClassExp (*i.e.*, calling its eval method) returns a Val object. Looking only at the syntax, it seems reasonable that such an object has a superclass (the ext [for EXTENDS] part), a set of static identifiers and associated values, a set of field names, and a set of method names and associated procedures.

We define a StdClass object in the file class, which extends the Val class. (Actually, it extends ClassVal class which in turn extends the Val class.) Evaluating a ClassExp expression returns an instance of StdClass.

[Warning: In this chapter we implement classes and objects in our defined language (OBJ), using classes and objects in our defining language (Java). This can be confusing. For example, a ClassExp evaluates to an object in Java that represents a class in our defined language. Be sure that you have a clear understanding of the context in which the terms "class" and "object" are being used in the following discussion.]

A StdClass object (we're in the defining language, Java, now) has the following instance variables:

```
public ClassVal superClass;
public Bindings staticBindings;
public Fields fields;
public Methods methods;
public Env staticEnv;
```

The superclass variable is a reference to a ClassVal object (defining) that is the superclass (defined) of this class (defined). The staticBindings variable is a reference to the the bindings of the static variable names (defined) to their RHS values (defined). The RHS values are evaluated in the current static environment. (More about this later.) [OK, so I'll stop the "(defined)" and "(defining)" stuff now, but *pay attention*.]

The fields and methods variables are references to the Fields and Methods objects that capture the class field names and method names and procedures – these are not evaluated yet.

The staticEnv variable is a reference to the static environment of this class: it starts out extending the static environment of the superclass with an empty Bindings object named staticBindings. New bindings are added to the staticBindings list using the variable definitions in the statics parameter of the StdClass constructor. Each static LHS identifier is bound to the value of its RHS, where the RHS expression is evaluated in the current static environment. These bindings are created in order (first to last) as in top-level defines.

Two predefined identifiers are inserted initially into the list of static bindings: one binds the identifier myclass to (a reference to) this class itself, and another binds the identifier superclass to (a reference to) the superclass of this class.

If the class expression does not have an extends part, its superclass defaults to an unnamed "parentless" class (an EnvClass object in the Java implementation) whose static environment is is the top-level program environment — so top-level variables that have been defined are visible. In this way, all of the RHS expressions in the static definitions of such a class have access to the other static bindings in the class and to the top-level bindings.

A class expresion that specifies an explicit superclass has an environment that extends the static environment of that superclass, and – going up the superclass chain – has access to the bindings in all of the superclass static environments and the top-level bindings.

If a class expression has a static definition for a variable that also appears as a static variable in a superclass, that definition shadows the superclass variable. We disallow duplicate LHS variable names in a given class expression's static definitions, including redefinitions for myclass or superclass, although our implementation does not check for this.

Although counter-intuitive, objects are actually simpler than classes, because an object is essentially an environment!

An ObjectVal is a Java class that extends the Val class. It has a single instance variable:

public Env objectEnv;

The new operator in our defined language takes a class expression and returns a Java ObjectVal instance that essentially couples the static bindings of the class with bindings for the class fields and methods. Since our language does not define an explicit constructor in class expressions, object fields are initialized to nil. The method variable names in the class definition are bound to procedure closures that capture the environment that includes bindings for the class static variables (from the staticBindings field), along with bindings for the fields as described above. The method closures are created as in letrec, so they can refer to themselves recursively. As with static variables, we disallow duplicate method variable names, although our implementation does not check for this.

Before an object builds the environment that is specific to the fields and methods of the class, it creates an instance of the superclass of the class and adopts the environment of the superclass object (bound to the variable super) before adding static, field, and method bindings. Since creating the superclass object may itself involve creating an instance of *its* superclass, object creation continues up the class hierarchy until a parentless class is found, at which point there is no further super object to create.

At the top of the chain of super objects, the identifier self is bound to a reference to the object being created. In this way, the object can refer to itself through the self identifier. (In Java, we call it this instead of self.) Methods declared in superclasses that refer to self will "see" the original object, allowing for dynamic dispatch of method calls, an important feature of object-oriented languages. We call this a *deep* binding.

As objects are created up the superclass chain, the identifier this is bound to the object created by that class. We call this a *shallow* binding.

Notice that an object can see all of the static bindings up the superclass chain, but that if a static variable is bound to a procedure (or some other value that captures an environment), the procedure captures only the static environment of the class and cannot "see" any of the fields or methods – including the self and this identifiers – in its environment.

A parentless class is an instance of the Java EnvClass class, which has one instance variable:

```
public Env staticEnv;
```

Here is the code for the constructor, which is called when processing the <ext>:Ext0 grammar rule.

```
public EnvClass(Env env) {
    // the static environment of this class
    staticEnv = env;
}
```

As noted earlier, the static environment of a parentless class is the top-level program environment, which is passed into the Env env parameter when the class is created. (See the Ext0 class for details.)

A standard class (an instance of the Java StdClass class) has a similar constructor, except that it builds on the environment of its superclass as described earlier. A standard class defines bindings for the variables myclass and superclass, whose values are self-explanatory. You can find the code for this constructor in the file class.

As described on slides 9 and 10, as an object is created, the self identifier is bound to a reference to the object in the makeObject method in the parentless class. But how can this binding take place when the object has not yet been completely created? We do this by creating a Java ValRef object named objRef, with a dummy initial value (nil, to be precise), and pass this to the makeObject method.

The objRef is passed up the superclass chain through successive makeObject calls. When makeObject reaches a parentless class, the self identifier is put into the object environment, bound to objRef. After the original object has been completely created – that is, after all of the makeObject calls have returned, objRef is rebound to the newly created object with a call to setRef. Here is the complete eval code for a new expression in the NewExp part of the code file:

```
public Val eval(Env env) {
    // get the class from which this object is created
    Val val = exp.eval(env);
    // create a reference to a dummy value (nil)
    Ref objRef = new ValRef(Val.nil);
    // let the class create the object
    ObjectVal objVal = val.makeObject(objRef);
    // set the reference to the newly created object
    return objRef.setRef(objVal);
}
```

The makeObject method is defined for both a StdClass and an EnvClass. (For all other Val objects, makeObject throws an exception.)

In StdClass, makeObject first creates an instance of the superclass, and then stitches together an environment that includes the static bindings, the fields, and the methods. Three fields are created and initialized automatically: self is bound to the base object (a deep binding), super is bound to the instance of the superclass, and this is bound to this object (a shallow binding). The remaining named fields are bound to nil, and the methods are bound to closures as in letrec. The code for makeObject is given on the next two slides.

```
public ObjectVal makeObject(Ref objRef) {
    // create the parent object (recursively)
    ObjectVal parent = superClass.makeObject(objRef);
    // this object's environment extends the parent object's env
    Env env = parent.objectEnv;
    // add this class's static bindings (including myclass, etc.)
    env = env.extendEnvRef(staticBindings);
    // bind 'super' to the parent object
    // bind 'self' to the base object (deep)
    // bind 'this' to this object (shallow)
    // 'self' is unnecessary here, except that it speeds up lookups
    Bindings fieldBindings = new Bindings();
    env = env.extendEnvRef(fieldBindings);
    fieldBindings.add("super", new ValRef(parent)); // parent object
    fieldBindings.add("self", objRef); // deep
    fieldBindings.add("this", new ValRef(objectVal)); // shallow
    // next bind all of this object's instance fields to nil
    for (Token t : fields.varList) {
        String s = t.toString();
        fieldBindings.add(s, new ValRef(Val.nil));
    env = env.extendEnvRef(fieldBindings);
```

... continued on next slide ...

... continued from previous slide ...

```
// bind all this object's methods as in letrec
if (methods.varList.size() > 0) {
    LetrecDecls methodDecls =
        new LetrecDecls(methods.varList, methods.procList);
    env = methodDecls.addBindings(env);
}

// create the object and return it
return new ObjectVal(env);
}
```

The makeObject method for EnvClass is simple, since it's always the last superclass object that needs to be constructed. It extends the environment in which the EnvClass is created (*not* the top-level environment) with a single field binding of self to (a reference to) the object being created.

```
public ObjectVal makeObject(Ref objRef) {
    // start with the static environment of this class
    Env env = staticEnv;
    // add the field binding 'self' to refer to this object
    Bindings fieldBindings = new Bindings();
    fieldBindings.add("self", objRef);
    env = env.extendEnvRef(fieldBindings);
    return new ObjectVal(env);
}
```

Such an ObjectVal can access the static environment of this class, which is the top-level program environment.

Observe that the RHS expressions in static definitions are evaluated in the static environment of the class, which extends the static environment of its superclass (and so on...). For example, consider the following expression:

```
define c = class static x = 5 end
define x = 3
define cc =
  class extends c
    static y = x
  end
```

In this case, the variable y in class cc is bound to 5, not 3.

The < . . . > operator, when applied to a class, extracts the static environment of the class. When used in an expression of the form

the expression exp is evaluated in the static environment of the given class. This can be used to evaluate an expression that refers to any static variable in the class.

Consider a class c, for example. For a static variable x in the class c, the expression <c>x evaluates to the value of the variable. For a static procedure f in the class c, the expression .<c>f (...) evaluates to the application of f to its actual parameters.

When making a procedure application such as .<c>f(...), it is important to know

- the environment in which f is evaluated, and
- the environment in which the actual parameters to f are evaluated.

To evaluate the expression .<c>f(...), we evaluate the procedure f in the environment determined by <c>, whereas we evaluate the actual parameter expressions in the environment in which the application .<c>f(...) is made. For example, in

```
define f = proc(t) *(2,t)
define c =
    class
        static x = 3
        static f = proc(t) t
    end
let
    x = 5
in
    .<c>f(x)
```

the expression .<c>f(x) evaluates to 5, because f is bound to the procedure given in the static definition of c, but its actual parameter x evaluates to 5, since the actual parameter expression is evaluated in the local let environment. In other words, .<c>f(x) is the same as $.{<c>f}(x)$.

Keep in mind that the following two expressions are not equivalent:

```
.<c>f(...)
<c>.f(...)
```

In the first expression, f is evaluated in the static environment of c. Then the procedure bound to f is applied to the actual parameters (...) which are evaluated in the current environment, *not* in the static environment of c.

In the second expression, the entire expression f(...) is evaluated in the static environment of f(...) are also evaluated in this static environment.

In the example on the previous slide, if the final expression was <c>.f(x) instead of .<c>f(x), it would evaluate to 3, since x is bound to 3 in the static environment of c.

To clarify, the above two expressions can be re-written as follows:

```
.{<c>f}(...)
<c>{.f(...)}
```

The static environment of a class ultimately ends up extending the top-level program environment, not the environment in which the class is defined. Consider the following code:

```
define x = 3
let
  x = 5
in
  <class end>x
```

In this example, the class is defined in the let environment, but its static environment extends the top-level environment, not the let environment, so the value of this expression is 3, not 5.

There are situations in which we may want to retrieve the value of a variable in the "local" environment in which the class is defined and not in the static environment of the class. To do so, we predefine a static "variable" ! @ (called "bang-at") in every class and bind it to an object that captures the (local) environment in which the class is defined. This binding becomes part of the static environment of the class. The token '! @' is not really a variable, so it cannot appear in the LHS of an assignment. Its principal use is in expressions of the form

```
<!@>exp
```

which evaluates to the value of the expression exp in the local environment.

For example, consider again the expression

```
define x = 3
let
  x = 5
in
  <class end>x
```

where we observed that this expression evaluates to 3. If we replace

```
<class end>x
```

in the above expression with

```
<class end><!@>x
```

then this expression evaluates to 5 because the class expression is defined in the let environment, and so the environment captured by !@ has x bound to 5.

Observe that the expression ! @ is only meaningful in the context of the static environment of a class. If used anywhere else, it will throw an "unbound variable" exception.

The special operator @ returns an object that captures the *current* environment, whatever that may be. (Recall your homework assignment that introduced this notation.) This operator may be used to pass a captured environment as an argument to a procedure application or to assign it to a variable for later reference.

```
AtExp
%%%

   public Val eval(Env env) {
      return new ObjectVal(env);
}
```

The special operator @@ does the same thing, except that it also displays the current environment in a human-readable way.

Notice that @ is meaningful in *any* expression context – because every expression is evaluated in *some* environment – but that ! @ is only meaningful in an expression that appears in the context of a class, and its value represents an environment entirely separate from the static environment of the class.

Some examples:

```
define x = 11
define y = 42
define z = 666
define xyenv =
 let
 x = 3
 y = 5
 in
   (d
<@>x % => 11
<@>y % => 42
<@>z % => 666
<xyenv>x % => 3
<xyenv>y % => 5
< xyenv > z % => 666 (the 'let' extends the top-level env)
```

Observe that for any expression exp, the following two expressions evaluate to the same values:

Unlike the new operator in Java, Our new operator does not take any arguments, and all of the fields are initialized to nil. We can, of course, initialize fields by calling a method. Here's an example:

```
let
  c = class
      field x
      field y
      method init = proc(a,b) { set x=a; set y=b; self }
      end
  in
    let
      o = .<new c>init(3,4)
    in
      <o>+(x,y) % => 7
```

Since the init method returns self, the value of the expression .<new c>init (3, 4) is the *same* object as the one created by the expression new c, except that its fields x and y are set to values 3 and 4, respectively.

You might be inclined to think that .<new c>init(a,b) is the same as <new c>{ set x=a ; set y=b ; self}, but the bindings for a and b may be different in these two expresions.

In the previous example, the init method is invoked separately, after the object is created using new, and not as part of the object creation itself. Naming this procedure "init" is not a requirement. A class can have several methods that initialize its fields, much as a Java class can have several constructors. Unlike Java, the OBJ language can apply its methods – even the ones intended to initialize the fields – at any time.

The OBJ language has three additional expressions, with the following grammar rules:

The DISPLAY, DISPLAY1, and NEWLINE tokens are defined by

```
DISPLAY 'display'
DISPLAY1 'display#'
NEWLINE 'newline'
```

Evaluating a DisplayExp expression results in evaluation of its <exp> part in the current environment; this value's toString() representation is then displayed on standard output, and the value is returned as the value of the DisplayExp expression. Display1Exp is like DisplayExp except that the displayed value is followed by a single space. The value of a NewlineExp expression is nil, and a newline is displayed on standard output.

5.30

Language OBJ (continued)

Here are examples of how to use display, display#, and newline:

```
let
  x = 3
  y = 5
  z = 8
in
  { display x ; newline
  ; display y ; newline
  ; display z ; newline
  ; nil
  }
```

Evaluating this expression results in displaying the following to standard output (omitting the final nil):

3

5

8

If the newline expressions are removed, the output appears as follows (omitting the final nil):

358

If display is then replaced by display#, the output appears as follows:

3 5 8

Consider the following OBJ program:

```
define summer =
  class
    field sum
  method init = proc() {set sum = 0 ; self}
  method add = proc(t) {set sum = +(sum,t) ; self}
  method show = proc() {display sum ; newline ; self}
  end
```

Here's an example of how this class might be used to find and display the sum of the integers 1, 3, 5, and 7:

```
define o = .<new summer>init()
.<o>add(1)
.<o>add(3)
.<o>add(5)
.<o>add(7)
.<o>show()
```

Since init and add both return self, the following "one-liner" would accomplish the same thing:

```
.<..<..<..<new summer>init()>add(1)>add(3)>add(5)>add(7)>show()
```

Since we often find ourselves encountering expressions like the following:

```
.<..<..<new summer>init()>add(1)>add(3)>add(5)>add(7)>show()
we introduce a short-hand way of writing this:
```

```
!<new summer>init()>add(1)>add(3)>add(5)>add(7)>show()!>
```

The token LLANGLE, defined as the string '!<', introduces an expression — which must evaluate to an environment (class or object) — followed by a sequence of zero or more procedure applications, each preceded by '>'. Each procedure is evaluated in the environment of the previous class or object, and the procedure application itself must return another object, whose environment is then used to evaluate the next procedure, and so on. The entire expression is terminated by a RRANGLE token, defined as '!>'. Note that the actual parameter expressions are evaluated in the environment in which the entire expression appears.

Here are the associated grammar rules:

The BNF identifier mangle should suggest "multiple angle (brackets)", but it also could also appropriately be interpreted as a twisted mess.

Here is the implementation of how to evaluate an EenvExp expression.

The expression exp is evaluated in the current environment. Its value v, along with the current environment, is passed to the mangle object, which then evaluates the subsequent procedure applications.

Mangle(List<Exp> expList, List<Rands> randsList)

A mangle object consists of a list of Exp objects and a list of Rands objects.

Here is the code for eval (Val v, Env env) in the Mangle class:

```
public Val eval(Val v, Env env) {
    Iterator<Exp> expIter = expList.iterator();
    Iterator<Rands> randsIter = randsList.iterator();
    while (expIter.hasNext()) {
        // expIter.next() ProcExp to apply
        // v.env() is the environment in which to build the ProcVal
        Val p = expIter.next().eval(v.env());
        // evaluate this method's rands in env
        List<Val> valList = randsIter.next().evalRands(env);
        v = p.apply(valList);
    }
    return v;
}
```

Each Exp is evaluated in the environment defined by the value v (which must, perforce, be a class or object), and this must evaluate to a ProcVal (named p in this code). The valList arguments to p are evaluated in the outside environment env (not in the environment defined by v) from the corresponding Rands object. The procedure p is then applied to these arguments, and the resulting application becomes the new v. This is repeated until all of the Mangle applications are performed. The final value v is returned as the result of the EenvExp expression.

It turns out that exposing the environment of a procedure can be used to implement, with procedures alone, a simplified approach to objects and methods. See the file

for an example. On the other hand, exposing the environment of a procedure can result in modifying that environment, which can lead to unintended consequences.

In the OBJ language, the send operator provides syntactic sugar that is specific to method calls on a object. Specifically, for an object o and method m, the following expressions are equivalent:

```
send o m(...)
.<o>m(...)
```

You can think of a send expression such as above as sending a message m to the object o. Some examples appear on the following slide.

```
define c1 = class
 field x field y
 method init1 = proc(v, w) {set x=v; set y=w}
 method getx1 = proc() x
 method gety1 = proc() y
end
define c2 = class extends c1
 field y
 method init2 = proc(w) set y=w
 method getx2 = proc() x
 method gety2 = proc() y
end
define o2 = new c2
send o2 init1(1,2) % same as .<o2>init1(1,2)
send o2 init2(9) % same as .<o2>init2(9)
send o2 getx1() % same as .<o2>getx1() => 1
send o2 gety1() % same as .<o2>gety1() => 2
send o2 getx2() % same as .<o2>getx2() => 1
send o2 qety2() % same as .<o2>qety2() => 9
```

In method application, self always refers to the base object, even if self appears in the definition of a superclass method. This is what makes dynamic dispatch work!

```
define c1 =
 class
   method m1 = proc() 1
    method m2 = proc() send self m1()
 end
define c2 =
 class extends cl
   method m1 = proc() 2
 end
define o1 = new c1
define o2 = new c2
send o1 m1() % same as .<o1>m1() => 1
send o2 m1() % same as .<o2>m1() => 2
send o2 m2() % same as .<o2>m2() => 2!
```

The following slide gives another example of dynamic dispatch.

```
define shape =
  class
    method area = proc() 0 % shapeless
  end
define rectangle =
  class extends shape
   field len % length
   field wid % width
    method init = proc(len,wid) {set <this>len=len; set <this>wid=wid; self}
   method area = proc() * (len, wid)
  end
define circle =
  class extends shape
    field rad % radius
    method init = proc(rad) {set <this>rad=rad; self}
    method area = proc() *(3,*(rad,rad)) % a bit of an underestimate
  end
define r = . < \text{new rectangle} > \text{init}(4,5) % a rectangle with length 4 and width 5
define c = .< new circle > init(2) % a circle with radius 2
define s = new shape
.<r>area() % => 20
.<c>area() % => 12
. < s > area() % => 0
```

Other examples on this slide and the next ...

```
define c1 =
 class
    method m1 = proc() send self m2()
   method m2 = proc() 13
 end
define c2 =
  class extends c1
   method m1 = proc() 22
    method m2 = proc() 23
    method m3 = proc() send super m1()
 end
define c3 =
  class extends c2
    method m1 = proc() 32
   method m2 = proc() 33
 end
define o3 = new c3
send o3 m3() % returns 33
```

```
define a = class
    field i field j
   method setup = proc() {set i=15; set j=20; 50}
   method f = proc() .<self>q()
   method g = proc() + (i, j)
  end
define b = class extends a
   field j field k
   method setup =
      proc() {set j=100; set k=200; .<super>setup(); .<self>h()}
   method q = proc() [i, j, k]
   method h = proc() .<super>q()
  end
define c = class extends b
   method g = proc() .<super>h()
   method h = proc() + (j,k)
  end
let
  p = proc(0)
     let.
     u = .<o>setup()
     in
       [u, .<o>q(), .<o>f()]
in
  [.p(new a), .p(new b), .p(new c)]
% returns [[50,35,35],[35,[15,100,200],[15,100,200]],[300,35,35]]
```

Language PROP

In many object-oriented programming languages, the fields of an object can be made *private* – that is, inaccessible outside of the object's methods. For private fields, special publically accessible methods can be used to retrieve the field values or to modify them. These methods are often called *getters* and *setters*, respectively.

Suppose, for example, we provided a special designator called private that served to identify a field whose value was inaccessible outside of the class methods. Consider the following code:

```
define c =
  class
    private x
    method get_x = proc() x
    method set_x = proc(v) set x = v
    end
define cc = new c
.<cc>set_x(5) % ok - sets value of x to 5
.<cc>get_x() % ok - returns 5
<cc>x
    % illegal - x is private
```

While this sort of code is common, there are two problems with this approach. The first is that every private field we want to access needs a getter and a setter. The second is that code such as <cc>set x = 5 does not work, but is easier to write and understand than $.<cc>set_x(5)$.

The C# language championed by Microsoft solves these problems using the notion of a *property*. A property acts like a field but it provides built-in getter and setter code. When the field is *accesssed*, the getter code is executed; when the field is *assigned to* with a set statement, the setter code is executed.

Here is the same class as described on the previous slide, with a property instead of a getter and setter:

```
define c =
  class
    field x
    property x = prop x:set x=$
  end
define cc = new c
<cc>set x = 5 % ok - sets the field value to 5
<cc>x
    % returns 5
```

The property x shadows the field x. This means that the field x cannot be accessed directly except through the property.

Here are the grammar rules for defining properties in a class defintion:

When a variable bound to a property is evaluated, its getExp code is executed using the environment captured where the property is defined, and the expressed value of the variable is the result of evaluating the getExp expression.

When a variable bound to a property is assigned to in a set expression, the RHS of the expression is evaluated in the current environment. The environment where the property is defined is then extended by binding the special symbol '\$' to the value of the RHS. The property's setExp expression is then evaluated in this extended environment, and the resulting value is the value of the setExp expression.

If a variable z [for example] is bound to a property whose set expression has no side-effects [such as nil], an expression such as set z = ... does not modify anything – including z.

```
define c =
  class
    field x
    method init = proc(x) {set <this>x = x; self}
    property y = prop x:set x=$
    property z = prop x:nil
  end
define o = .< new c > init(3)
\langle o \rangle [x,y,z] % [3,3,3]
<o>set x = 5
\langle o \rangle [x, y, z] % [5, 5, 5]
<o>set y = 11
\langle o \rangle [x, y, z]  % [11, 11, 11]
<o>set z = 42
\langle o \rangle [x, y, z] % [11, 11, 11]
```

We implement properties in the same way we implement call-by-name: a property evalutes to a thunk-like object called a PropRef (which extends the Ref class); this object captures the environment in which the property is defined, along with the property's get and set expressions.

The expressed value of a variable bound to a property is the value obtained by evaluating the property's get expression in the captured environment by calling the thunk's deRef method.

Similarly, when assigning a value to a variable bound to a property, the thunk's setRef method is called, with the assigned value bound to the special variable '\$' and returning the value of the property's set expression.

If a field named x in a class has a property also named x, referring to x in the contect of this object uses the PropRef instead of the variable. Consider this example:

```
define c =
  class
    static p = proc(t) set t=add1(t)
    field x
    method init = proc(x) {set <this>x=x; self}
    property x = prop x : set x=$
    property y = prop x : set x=+($,$)
    end
define o = .<new c>init(3)
<o>{.p(x); x} % evaluates to 4
<o>{.p(y); x} % evaluates to 10
```

In the expression .p(x), x refers to the property x. Since we are using call-by-reference parameter passing semantics, the variable t is bound to this property, and the expression set t=add1(t) is evaluated using this binding. Remember that the RHS parts of property definitions are all evaluated in the environment that includes only statics, fields, and methods, but not properties.

So far, a property is only defined in the conext of a class definition, where it plays a role in object instantiantion. It turns out that the behavior of properties could be useful even outside of the context of an object, especially to manage access to variables defined in a let expression. To make this explicit, we create a letprop construct that has the following concrete syntax and abstract class structure:

Consider this (somewhat strange) example:

```
let
    x = 3
in
    letprop
    x = prop x : set x = 5
    in
    {set x = 42 ; x} % => 5
```

This expression evaluates to 5.

The set part of a prop is optional. If the set part is omitted, any attempt to apply the set operator to the variable results in a runtime exception. In this way, we can implement "read-only" properties.

```
let
  x = 3
in
  letprop
  x = prop x %% no 'set' part, so x is read only
in
  {set x = 42; x} % => runtime exception
```

It turns out that a read-only prop behaves just like a thunk in call-by-name (which, you may recall, is also read-only), so we have the benefit of call-by-name semantics when we want it!

```
let
  while = proc(test?, do, ans)
    letrec
      loop = proc()
        if test? then {do ; .loop()} else ans
    in
      .loop()
  sum = 0
 count = 10
  i = 1
in
  letprop
    test? = prop count
    do = prop \{ set sum = +(sum, i) \}
              ; set i = add1(i)
              ; set count = sub1(count)
    ans = prop sum
  in
    .while(test?, do, ans) % => 55 = 1+2+...+10
```